
Subject: Re: How 2 show animations.
Posted by David Fanning on Fri, 02 Mar 2012 14:15:00 GMT
View Forum Message <> Reply to Message

DavePoreh writes: 

>  I have 52 images (in .png) format and i would like to make it like
>  animation in IDL. can anyone help me.

If you have IDL 8, the IDLffVideoWrite object produces
animations in mpg4 and avi format that don't look so
bad. You can specify which Codec you want to use to
build the animation, and so on. Quite a few features.

Here is a very simple example that makes an AVI movie of all
the PNG files in the current directory. Presumably, you
have named them in a sensible way. :-)

;--------------------------------------------------------
PRO Movie_Example

  Compile_Opt idl2
  
  ; Open the video recorder.
  video_file = 'movie_example.avi'
  video = IDLffVideoWrite(video_file, Format='avi')
  
  ; Configure the video output for the PNG files you
  ; plan to add.
  framerate = 5
  xsize = 750
  ysize = 600
  stream = video.AddVideoStream(xsize, ysize, framerate)
  
  ; Get the files.
  files = File_Search('*.png', COUNT=fileCnt)
  FOR j=0,fileCnt-1 DO BEGIN
     
    ; Read the PNG file.
    image = Read_PNG(files[j])
    
    ; Save the image in the video stream
    void = video.Put(stream, image)
  
  ENDFOR
  
  ; Clean things up.
  video.Cleanup

END
;--------------------------------------------------------

Cheers,

David

-- 
David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: http://www.idlcoyote.com/
Sepore ma de ni thui. ("Perhaps thou speakest truth.")

